QHD-2, QHD-3 and QHD-4 decay from a metastable potential

by José Luis Gómez-Muñoz
http://homepage.cem.itesm.mx/lgomez/quantum/
jose.luis.gomez@itesm.mx

Introduction

Higher Order Quantized Hamilton Dynamics (QHD) is applied to the **decay from a metastable potential**. QHD gives an approximation to the Heisenberg Equations of Motion (EOM). The QHD commands used in this document are included in QUANTUM, which is a free *Mathematica* add-on that can be downloaded from
http://homepage.cem.itesm.mx/lgomez/quantum/
This tutorial shows how to use the QUANTUM *Mathematica* add-on to reproduce the results and graphs from Pahl and Prezhdo in J. Chem Phys. Vol 116 No. 20, May 2002, Pages 8704-8712
http://homepage.cem.itesm.mx/lgomez/quantum/QHDHigherOrders.pdf

Load the Package

First load the Quantum`QHD` package. Write:
Needs["Quantum`QHD`"]
then press at the same time the keys SHIFT-ENTER to evaluate. *Mathematica* will load the package and print a welcome message:

```
Needs["Quantum`QHD`"]
```

```
Quantum`QHD`
A Mathematica package for Quantized Hamilton
  Dynamics approximation to Heisenberg Equations of Motion
by José Luis Gómez-Muñoz
based on the original idea of Kirill Igumenshchev

This add-on does NOT work properly with the debugger turned on. Therefore
  the debugger must NOT be checked in the Evaluation menu of Mathematica.

Execute SetQHDAliases[] in order to use the keyboard to enter QHD objects
SetQHDAliases[] must be executed again in each new notebook that is created
```

In order to use the keyboard to enter quantum objects write:
SetQHDAliases[ ];
then press at the same time the keys SHIFT-ENTER to evaluate. Remember that SetQuantumAliases[ ] must be evaluated again in each new notebook:

```
SetQHDAliases[]
```

```
ALIASES:
[ESC]on[ESC]      · Quantum concatenation symbol
[ESC]time[ESC]    t Time symbol
[ESC]hb[ESC]      ℏ Reduced Planck's constant (h bar)
[ESC]ii[ESC]      ⅈ Imaginary I symbol
[ESC]inf[ESC]     ∞ Infinity symbol
[ESC]->[ESC]      → Option (Rule) symbol
[ESC]ave[ESC]     〈□〉 Quantum average template
[ESC]expec[ESC]   〈□〉 Quantum average template
[ESC]symm[ESC]    (□·□)_S Symmetrized quantum product template
[ESC]comm[ESC]    ⟦□,□⟧_ Commutator template
[ESC]po[ESC]      (□)^□ Power template
[ESC]su[ESC]      □_□ Subscripted variable template
[ESC]posu[ESC]    □_□^□ Power of a subscripted variable template
[ESC]fra[ESC]     □/□ Fraction template
[ESC]eva[ESC]     □/.{□→□,□→□} Evaluation (ReplaceAll) template

SetQHDAliases[] must be executed again in each
  new notebook that is created, only one time per notebook.
```

Commutation Relationships and Hamiltonian

This Hamiltoninan and commutation relationships are those used by Pahl and Prezhdo in their paper.
In order to enter the templates and symbols $[\![\square, \square]\!]_-$, $\square_\square$, i and $\hbar$ you can either use the QHD palette (toolbar) or, if the command SetQHDAliases[] has been executed in this *Mathematica* notebook, then press the keys [ESC]comm[ESC], [ESC]su[ESC], [ESC]ii[ESC] and [ESC]hb[ESC] respectively:

SetQuantumObject[q, p];

$$[\![q, p]\!]_- = i * \hbar;$$

$$mh = \frac{p^2}{2\,m} + \frac{a_2}{2}\,q^2 + \frac{a_3}{3}\,q^3$$

$$\frac{p^2}{2\,m} + \frac{q^2\,a_2}{2} + \frac{q^3\,a_3}{3}$$

Heisenberg Equations of Motion

The evolution of the expected value of an observable A in the Heisenberg representation is given by the equation of motion (EOM):

$$i\,\hbar\,\frac{d}{dt}\,\langle A\rangle = \langle [A, H]\rangle$$

The Heisenberg EOM can be calculated using the QHD-*Mathematica* command QHDEOM, as shown below. The first argument specifies the QHD-order, which is set below to ∞ so that no QHD approximation is made in this first example. The second argument is the observable, which is set to *p* in the example below. The third argument is the Hamiltonian, remember that it was stored in variable *mh* above in this document. Finally options can be included, for example QHDHBar→1 specifies that reduced Planck's constant (Dirac constant) will take the value of 1 (The → symbol can be entered by pressing the keys [ESC]->[ESC], that is [ESC][MINUS][GREATERTHAN][ESC]). The output of QHDEOM can be stored in a variable; it is stored in *meom* below:

```
meom = QHDEOM[∞, p, mh]
```

```
{{QHDLabel, No closure was applied}, {⟨p⟩, -a₂ ⟨q⟩ - a₃ ⟨q²⟩}}
```

The output of QHDEOM can be shown in a nicer format using QHDForm. The output *meom* that was stored above is formated that way below:

```
QHDForm[meom]
```

No closure was applied
$\frac{d\langle p\rangle}{dt} = -a_2 \langle q\rangle - a_3 \langle q^2\rangle$

On the other hand, the command QHDDifferentialEquations formats the output of QHDEOM as a standard *Mathematica* equation, as those that can be part of the input of standard *Mathematica* commands like DSolve and NDSolve. The output *meom* that was stored above is formated that way below:

```
QHDDifferentialEquations[meom]
```

$\{\langle p\rangle'[t] == -a_2 \langle q\rangle[t] - a_3 \langle q^2\rangle[t]\}$

A different symbol for time can be specified. The operator → can be entered pressing the keys [ESC][MINUS][GREATERTHAN][ESC], see the equations below:

```
QHDDifferentialEquations[meom, QHDSymbolForTime → z]
```

$\{\langle p\rangle'[z] == -a_2 \langle q\rangle[z] - a_3 \langle q^2\rangle[z]\}$

The EOM for $\langle p\rangle$ depends on $\langle q\rangle$ and $\langle q^2\rangle$ (please remember that $\langle A^2\rangle$ is NOT the same as $\langle A\rangle^2$). Next, the EOMs of $\langle q\rangle$ and $\langle q^2\rangle$ can generate new expectation values, such as $\langle p \cdot q\rangle_s = \left\langle \frac{p\cdot q+q\cdot p}{2} \right\rangle$. Regarding each expectation value as a dynamical variable, we can generate a hierarchy of EOMs, which in general is an infinite hierarchy. We can obtain some of the EOM's of that infinite hierarchy using the QHD-*Mathematica* command QHDHierarchy, which has the same syntax as the command QHDEOM that was used above. Likewise, the output of QHDHierarchy can be stored in a variable (*mhie* in the example below) and shown using the command QHDForm below:

```
mhie = QHDHierarchy[∞, p, mh];
QHDForm[mhie]
```

Calculations were stopped at order QHDMaxOrder→2
$\frac{d\langle p\rangle}{dt} = -a_2 \langle q\rangle - a_3 \langle q^2\rangle$
$\frac{d\langle q\rangle}{dt} = \frac{\langle p\rangle}{m}$
$\frac{d\langle q^2\rangle}{dt} = \frac{2\langle pq\rangle_s}{m}$
$\frac{d\langle pq\rangle_s}{dt} = \frac{\langle p^2\rangle}{m} - a_2 \langle q^2\rangle - a_3 \langle q^3\rangle$
$\frac{d\langle p^2\rangle}{dt} = -2\, a_2 \langle pq\rangle_s - 2\, a_3 \langle pq^2\rangle_s$

A EOM hierarchy can be shown in a graph using the command QHDGraphPlot on the output of QHDHierachy. Arrows point from a first dynamical variable to a second dynamical variable that includes the first one in its EOM; please compare the table above with the graph below:

```
QHDGraphPlot[mhie]
```

Calculations were stopped at order QHDMaxOrder→2



QHD-2 Closure Approximation

QHDHierarchy stops calculating infinite hierarchies as specified by its option QHDMaxOrder, which takes a default value of 2. Some of the dynamical variables (expectation values) in the truncated hierarchy will not have their equation of motion (EOM) included, as it was shown above by the colors in the ouput of QHDGraphPlot. A closure procedure has to be applied in order to approximate those dynamical variables in terms of those other variables that do have their EOMs included in the hierarchy. For instance, the approximation

$\langle ABC\rangle \approx \langle AB\rangle\langle C\rangle + \langle AC\rangle\langle B\rangle + \langle BC\rangle\langle A\rangle - 2\langle A\rangle\langle B\rangle\langle C\rangle$

is used to approximate the third-order dynamical variables $\langle q^3\rangle$ and $\langle pq^2\rangle_s$ in terms of the firts and second order variables $\langle p\rangle$, $\langle q\rangle$, $\langle p^2\rangle$, $\langle q^2\rangle$ and $\langle p\cdot q\rangle_s = \left\langle \frac{p\cdot q+q\cdot p}{2}\right\rangle$, thus we obtain a second order QHD finite hierarchy of equations, QHD-2. The QHD-2 hierarchy is obtained as the output of the command QHDHierarchy with the first argument set to 2. Same as above, the second argument is the first variable in the hierarchy, the third argument is the hamiltonian, and optional arguments can be given after that. The output of QHDHierarchy is stored in the variable *mhier2*, and it is shown using the QHDForm command below:

```
mhier2 = QHDHierarchy[2, p, mh];
QHDForm[mhier2]
```

Closure procedure was applied to order 2
$\frac{d\langle p\rangle}{dt} = -a_2\langle q\rangle - a_3\langle q^2\rangle$
$\frac{d\langle q\rangle}{dt} = \frac{\langle p\rangle}{m}$
$\frac{d\langle q^2\rangle}{dt} = \frac{2\langle pq\rangle_s}{m}$
$\frac{d\langle pq\rangle_s}{dt} = \frac{\langle p^2\rangle}{m} + 2a_3\langle q\rangle^3 - a_2\langle q^2\rangle - 3a_3\langle q\rangle\langle q^2\rangle$
$\frac{d\langle p^2\rangle}{dt} = 4a_3\langle p\rangle\langle q\rangle^2 - 2a_3\langle p\rangle\langle q^2\rangle - 2a_2\langle pq\rangle_s - 4a_3\langle q\rangle\langle pq\rangle_s$

The closed QHD-2 hierarchy can be shown using the QHDGraphPlot command, compare the table above and the graph below:

```
QHDGraphPlot[mhier2]
```

Closure procedure was applied to order 2



Solution of the QHD-2 Equations: Decay from Metastable Potential

Next we evaluate the hierarchy when the parameters tkes the numerical values used by Pahl and Prezhdo in their paper, and the result of that evaluation is stored in the variable *mnumhier2*:

```
mnumhier2 = mhier2 /. {m → 1, a2 → 1, a3 → 1/2, ℏ → 1};
QHDForm[mnumhier2]
```

Closure procedure was applied to order 2
$\frac{d\langle p\rangle}{dt} = -\langle q\rangle - \frac{\langle q^2\rangle}{2}$
$\frac{d\langle q\rangle}{dt} = \langle p\rangle$
$\frac{d\langle q^2\rangle}{dt} = 2\,\langle pq\rangle_s$
$\frac{d\langle pq\rangle_s}{dt} = \left\langle p^2\right\rangle + \langle q\rangle^3 - \left\langle q^2\right\rangle - \frac{3}{2}\,\langle q\rangle\,\left\langle q^2\right\rangle$
$\frac{d\langle p^2\rangle}{dt} = 2\,\langle p\rangle\,\langle q\rangle^2 - \langle p\rangle\,\left\langle q^2\right\rangle - 2\,\langle pq\rangle_s - 2\,\langle q\rangle\,\langle pq\rangle_s$

The command QHDInitialConditionsTemplate generates an initial conditions template for the hierarchy:

```
QHDInitialConditionsTemplate[mnumhier2, 0]
```

$\{\langle p\rangle[0] == ■, \langle q\rangle[0] == ■, \langle q^2\rangle[0] == ■, \langle (p\cdot q)_s\rangle[0] == ■, \langle p^2\rangle[0] == ■\}$

Copy-paste the output of the previous command as input in the next one. Fill in the placeholders (■) with the appropiate initial values. Those initial values can be numbers. On the other hand, in the calculation below they are symbols like *po*, and then these symbols are evaluated at the desired numerical values. These initial conditions correspond to a Gaussian wavepacket, J. Chem Phys. Vol 113 No. 20, May 2002, Pages 8704-8712
http://homepage.cem.itesm.mx/lgomez/quantum/QHDHigherOrders.pdf
The initial conditions are stored in the variable *minicond2* below:

$$\text{minicond2} = \left\{\langle p\rangle[0] == po, \langle q\rangle[0] == qo, \langle q^2\rangle[0] == qo^2 + \frac{\hbar}{2\, m * \omega}, \langle (p\cdot q)_s\rangle[0] == po * qo, \langle p^2\rangle[0] == po^2 + \frac{\hbar * m * \omega}{2}\right\} /. \{qo \to 0, po \to 0, \omega \to 1, m \to 1, \hbar \to 1\}$$

$$\left\{\langle p\rangle[0] == 0, \langle q\rangle[0] == 0, \langle q^2\rangle[0] == \frac{1}{2}, \langle (p\cdot q)_s\rangle[0] == 0, \langle p^2\rangle[0] == \frac{1}{2}\right\}$$

The command QHDNDSolve takes as arguments the numerical version of the hierarchy (which was stored in the variable *mnumhier2*), the initial conditions (which were stored in *minicond2*), the initial time and the final time. The output of this command is the numerical solution of the differential equations, in the form of InterpolatingFunction objects. This output can be used to plot (graph) the dynamical variables as functions of time, as it will be shown below in this document. The output is stored in the variable *msol2* in the calculation below:

```
msol2 = QHDNDSolve[mnumhier2, minicond2, 0, 6.4]
```

$$\{\{\langle p\rangle \to \text{InterpolatingFunction}[\{\{0., 6.4\}\}, <>],$$
$$\langle q\rangle \to \text{InterpolatingFunction}[\{\{0., 6.4\}\}, <>],$$
$$\langle q^2\rangle \to \text{InterpolatingFunction}[\{\{0., 6.4\}\}, <>],$$
$$\langle (p\cdot q)_s\rangle \to \text{InterpolatingFunction}[\{\{0., 6.4\}\}, <>],$$
$$\langle p^2\rangle \to \text{InterpolatingFunction}[\{\{0., 6.4\}\}, <>]\}\}$$

The command command QHDPlot takes as its second argument the output of QHDNDSolve, which was stored in the variable *msol2*. The first argument specifies the dynamical variables or expression that we want to plot (graph) as a function of time, see the plot below:

```
QHDPlot[q, msol2]
```



QHDPlot accepts the same Options as the standard *Mathematica* command Plot. Some of those options are used, and the plot stored in the variable *mgraph2*, it will be used later in this document, see the commands below:

```
mgraph2 =
 QHDPlot[q, msol2, PlotStyle → Directive[Thick, DotDashed, Darker[Magenta]]]
```



QHD-3 Closure Approximation

The QHD-3 hierarchy for this problem can be obtained with the first argument of QHDHierarchy set to 3. This calculation takes several seconds in a laptop computer, the result is stored in the variable *mhier3* below:

```
mhier3 = QHDHierarchy[3, p, mh];
QHDForm[mhier3]
```

Closure procedure was applied to order 3
$\frac{d\langle p\rangle}{dt} = -a_2 \langle q\rangle - a_3 \langle q^2\rangle$
$\frac{d\langle q\rangle}{dt} = \frac{\langle p\rangle}{m}$
$\frac{d\langle q^2\rangle}{dt} = \frac{2\langle pq\rangle_s}{m}$
$\frac{d\langle pq\rangle_s}{dt} = \frac{\langle p^2\rangle}{m} - a_2 \langle q^2\rangle - a_3 \langle q^3\rangle$
$\frac{d\langle p^2\rangle}{dt} = -2\, a_2 \langle pq\rangle_s - 2\, a_3 \langle pq^2\rangle_s$
$\frac{d\langle q^3\rangle}{dt} = \frac{3\langle pq^2\rangle_s}{m}$
$\frac{d\langle pq^2\rangle_s}{dt} = -6\, a_3 \langle q\rangle^4 + 12\, a_3 \langle q\rangle^2 \langle q^2\rangle - 3\, a_3 \langle q^2\rangle^2 - a_2 \langle q^3\rangle - 4\, a_3 \langle q\rangle \langle q^3\rangle + \frac{2\langle p^2 q\rangle_s}{m}$
$\frac{d\langle p^2 q\rangle_s}{dt} = \frac{\langle p^3\rangle}{m} - 12\, a_3 \langle p\rangle \langle q\rangle^3 + 12\, a_3 \langle p\rangle \langle q\rangle \langle q^2\rangle - 2\, a_3 \langle p\rangle \langle q^3\rangle + 12\, a_3 \langle q\rangle^2 \langle pq\rangle_s - 6\, a_3 \langle q^2\rangle \langle pq\rangle_s - 2\, a_2 \langle pq^2\rangle_s - 6\, a_3 \langle q\rangle \langle pq^2\rangle_s$
$\frac{d\langle p^3\rangle}{dt} = \frac{\hbar^2 a_3}{2} - 18\, a_3 \langle p\rangle^2 \langle q\rangle^2 + 6\, a_3 \langle p^2\rangle \langle q\rangle^2 + 6\, a_3 \langle p\rangle^2 \langle q^2\rangle - 3\, a_3 \langle p^2\rangle \langle q^2\rangle + 24\, a_3 \langle p\rangle \langle q\rangle \langle pq\rangle_s - 6\, a_3 \langle pq\rangle_s^2 - 3\, a_2 \langle p^2 q\rangle_s - 6\, a_3 \langle q\rangle \langle p^2 q\rangle_s - 6\, a_3 \langle p\rangle \langle pq^2\rangle_s$

The products above are symmetrized, for instance $\langle p \cdot q^2\rangle_s = \left\langle \frac{p\cdot q^2 + q^2\cdot p}{2} \right\rangle$

The QHD-3 hierarchy can be shown using the QHDGraphPlot command, compare the table above and the graph below:

```
QHDGraphPlot[mhier3]
```

Closure procedure was applied to order 3



Solution of the QHD-3 Equations

Next we evaluate the hierarchy when the parameters tkes the numerical values used by Pahl and Prezhdo in their paper, and the result of that evaluation is stored in the variable *mnumhier3*:

```
mnumhier3 = mhier3 /. {m → 1, a2 → 1, a3 → 1/2, ℏ → 1};
QHDForm[mnumhier3]
```

Closure procedure was applied to order 3
$\frac{d\langle p\rangle}{dt} = -\langle q\rangle - \frac{\langle q^2\rangle}{2}$
$\frac{d\langle q\rangle}{dt} = \langle p\rangle$
$\frac{d\langle q^2\rangle}{dt} = 2\langle pq\rangle_s$
$\frac{d\langle pq\rangle_s}{dt} = \langle p^2\rangle - \langle q^2\rangle - \frac{\langle q^3\rangle}{2}$
$\frac{d\langle p^2\rangle}{dt} = -2\langle pq\rangle_s - \langle pq^2\rangle_s$
$\frac{d\langle q^3\rangle}{dt} = 3\langle pq^2\rangle_s$
$\frac{d\langle pq^2\rangle_s}{dt} = -3\langle q\rangle^4 + 6\langle q\rangle^2\langle q^2\rangle - \frac{3\langle q^2\rangle^2}{2} - \langle q^3\rangle - 2\langle q\rangle\langle q^3\rangle + 2\langle p^2 q\rangle_s$
$\frac{d\langle p^2 q\rangle_s}{dt} = \langle p^3\rangle - 6\langle p\rangle\langle q\rangle^3 + 6\langle p\rangle\langle q\rangle\langle q^2\rangle - \langle p\rangle\langle q^3\rangle + 6\langle q\rangle^2\langle pq\rangle_s - 3\langle q^2\rangle\langle pq\rangle_s - 2\langle pq^2\rangle_s - 3\langle q\rangle\langle pq^2\rangle_s$
$\frac{d\langle p^3\rangle}{dt} = \frac{1}{4} - 9\langle p\rangle^2\langle q\rangle^2 + 3\langle p^2\rangle\langle q\rangle^2 + 3\langle p\rangle^2\langle q^2\rangle - \frac{3}{2}\langle p^2\rangle\langle q^2\rangle + 12\langle p\rangle\langle q\rangle\langle pq\rangle_s - 3\langle pq\rangle_s^2 - 3\langle p^2 q\rangle_s - 3\langle q\rangle\langle p^2 q\rangle_s - 3\langle p\rangle\langle pq^2\rangle_s$

The command QHDInitialConditionsTemplate generates an initial conditions template for the hierarchy:

```
QHDInitialConditionsTemplate[mnumhier3, 0]
```

$\{\langle p\rangle[0] == ■, \langle q\rangle[0] == ■, \langle q^2\rangle[0] == ■, \langle (p\cdot q)_s\rangle[0] == ■, \langle p^2\rangle[0] == ■, \langle q^3\rangle[0] == ■, \langle (p\cdot q^2)_s\rangle[0] == ■, \langle (p^2\cdot q)_s\rangle[0] == ■, \langle p^3\rangle[0] == ■\}$

Copy-paste the output of the previous command as input in the next one. Fill in the placeholders (■) with the appropiate initial values. Those initial values can be numbers. On the other hand, in the calculation below they are symbols like *po*, and then these symbols are evaluated at the desired numerical values. These initial conditions correspond to a Gaussian wavepacket, J. Chem Phys. Vol 113 No. 20, May 2002, Pages 8704-8712
http://homepage.cem.itesm.mx/lgomez/quantum/QHDHigherOrders.pdf
The initial conditions are stored in the variable *minicond3* below:

```
minicond3 =
```

$$\left\{\langle p\rangle[0] == po,\ \langle q\rangle[0] == qo,\ \left\langle q^2\right\rangle[0] == qo^2 + \frac{\hbar}{2\, m * \omega},\ \langle (p\cdot q)_s\rangle[0] == po * qo,\ \left\langle p^2\right\rangle[0] == po^2 + \frac{\hbar * m * \omega}{2},\ \left\langle q^3\right\rangle[0] == qo^3 + \frac{3}{2}\, qo\, \frac{\hbar}{m * \omega},\ \left\langle \left(p\cdot q^2\right)_s\right\rangle[0] == po * qo^2 + \frac{1}{2}\, po\, \frac{\hbar}{m*\omega},\ \left\langle \left(p^2\cdot q\right)_s\right\rangle[0] == po^2 * qo + \frac{1}{2}\, qo * \hbar * m * \omega,\ \left\langle p^3\right\rangle[0] == po^3 + \frac{3}{2}\, po * \hbar * m * \omega\right\} \,/.\, \{qo \to 0,\ po \to 0,\ \omega \to 1,\ m \to 1,\ \hbar \to 1\}$$

$$\left\{\langle p\rangle[0] == 0,\ \langle q\rangle[0] == 0,\ \left\langle q^2\right\rangle[0] == \frac{1}{2},\ \langle (p\cdot q)_s\rangle[0] == 0,\ \left\langle p^2\right\rangle[0] == \frac{1}{2},\ \left\langle q^3\right\rangle[0] == 0,\ \left\langle \left(p\cdot q^2\right)_s\right\rangle[0] == 0,\ \left\langle \left(p^2\cdot q\right)_s\right\rangle[0] == 0,\ \left\langle p^3\right\rangle[0] == 0\right\}$$

The command QHDNDSolve takes as arguments the numerical version of the hierarchy (which was stored in the variable *mnumhier3*), the initial conditions (which were stored in *minicond3*), the initial time and the final time. The output of this command is the numerical solution of the differential equations, in the form of InterpolatingFunction objects. This output can be used to plot (graph) the dynamical variables as functions of time, as it will be shown below in this document. The output is stored in the variable *msol3* in the calculation below:

```
msol3 = QHDNDSolve[mnumhier3, minicond3, 0, 4.6]
```

$$\begin{aligned}\{\{&\langle p\rangle \to \text{InterpolatingFunction}[\{\{0., 4.6\}\}, <>],\\ &\langle q\rangle \to \text{InterpolatingFunction}[\{\{0., 4.6\}\}, <>],\\ &\left\langle q^2\right\rangle \to \text{InterpolatingFunction}[\{\{0., 4.6\}\}, <>],\\ &\langle (p\cdot q)_s\rangle \to \text{InterpolatingFunction}[\{\{0., 4.6\}\}, <>],\\ &\left\langle p^2\right\rangle \to \text{InterpolatingFunction}[\{\{0., 4.6\}\}, <>],\\ &\left\langle q^3\right\rangle \to \text{InterpolatingFunction}[\{\{0., 4.6\}\}, <>],\\ &\left\langle \left(p\cdot q^2\right)_s\right\rangle \to \text{InterpolatingFunction}[\{\{0., 4.6\}\}, <>],\\ &\left\langle \left(p^2\cdot q\right)_s\right\rangle \to \text{InterpolatingFunction}[\{\{0., 4.6\}\}, <>],\\ &\left\langle p^3\right\rangle \to \text{InterpolatingFunction}[\{\{0., 4.6\}\}, <>]\}\}\end{aligned}$$

The command command QHDPlot takes as its second argument the output of QHDNDSolve, which was stored in the variable *msol3*. The first argument specifies the dynamical variables or expression that we want to plot (graph) as a function of time, see the plot below:

```
QHDPlot[q, msol3]
```



QHDPlot accepts the same Options as the standard *Mathematica* command Plot. Some of those options are used, and the plot stored in the variable *mgraph3*, it will be used later in this document, see the commands below:

```
mgraph3 = QHDPlot[q, msol3, PlotStyle → Directive[Thick, Dashed, Darker[Brown]]]
```



The standard *Mathematica* command Show can be used to compare *mgraph3*, which was stored above, and *mgraph2*, which was stored in a previous section. The paper by Pahl and Prezhdo shows that QHD-3 is a better approximation to Quantum Mechanics than QHD-2, see the graph below:

```
Show[mgraph2, mgraph3]
```



QHD-4 Closure Approximation

The QHD-4 hierarchy for this problem can be obtained with the first argument of QHDHierarchy set to 4. This calculation takes several minutes in a laptop computer; the result is stored in the variable *mhier4* below:

```
mhier4 = QHDHierarchy[4, p, mh];
QHDForm[mhier4]
```

Closure procedure was applied to order 4
$\frac{d\langle p\rangle}{dt} = -a_2\langle q\rangle - a_3\langle q^2\rangle$
$\frac{d\langle q\rangle}{dt} = \frac{\langle p\rangle}{m}$
$\frac{d\langle q^2\rangle}{dt} = \frac{2\langle pq\rangle_s}{m}$
$\frac{d\langle pq\rangle_s}{dt} = \frac{\langle p^2\rangle}{m} - a_2\langle q^2\rangle - a_3\langle q^3\rangle$
$\frac{d\langle p^2\rangle}{dt} = -2\,a_2\langle pq\rangle_s - 2\,a_3\langle pq^2\rangle_s$
$\frac{d\langle q^3\rangle}{dt} = \frac{3\langle pq^2\rangle_s}{m}$
$\frac{d\langle pq^2\rangle_s}{dt} = -a_2\langle q^3\rangle - a_3\langle q^4\rangle + \frac{2\langle p^2q\rangle_s}{m}$
$\frac{d\langle q^4\rangle}{dt} = \frac{4\langle pq^3\rangle_s}{m}$
$\frac{d\langle p^2q\rangle_s}{dt} = \frac{\langle p^3\rangle}{m} - 2\,a_2\langle pq^2\rangle_s - 2\,a_3\langle pq^3\rangle_s$
$\frac{d\langle pq^3\rangle_s}{dt} = \frac{3\hbar^2}{2m} + 24\,a_3\langle q\rangle^5 - 60\,a_3\langle q\rangle^3\langle q^2\rangle + 30\,a_3\langle q\rangle\langle q^2\rangle^2 + 20\,a_3\langle q\rangle^2\langle q^3\rangle - 10\,a_3\langle q^2\rangle\langle q^3\rangle - a_2\langle q^4\rangle - 5\,a_3\langle q\rangle\langle q^4\rangle + \frac{3\langle p^2q^2\rangle_s}{m}$
$\frac{d\langle p^3\rangle}{dt} = -\hbar^2\,a_3 - 3\,a_2\langle p^2q\rangle_s - 3\,a_3\langle p^2q^2\rangle_s$
$\frac{d\langle p^2q^2\rangle_s}{dt} = 48\,a_3\langle p\rangle\langle q\rangle^4 - 72\,a_3\langle p\rangle\langle q\rangle^2\langle q^2\rangle + 12\,a_3\langle p\rangle\langle q^2\rangle^2 + 16\,a_3\langle p\rangle\langle q\rangle\langle q^3\rangle - 2\,a_3\langle p\rangle\langle q^4\rangle - 48\,a_3\langle q\rangle^3\langle pq\rangle_s + 48\,a_3\langle q\rangle\langle q^2\rangle\langle pq\rangle_s - 8\,a_3\langle q^3\rangle\langle pq\rangle_s + \frac{2\langle p^3q\rangle_s}{m} + 24\,a_3\langle q\rangle^2\langle pq^2\rangle_s - 12\,a_3\langle q^2\rangle\langle pq^2\rangle_s - 2\,a_2\langle pq^3\rangle_s - 8\,a_3\langle q\rangle\langle pq^3\rangle_s$
$\frac{d\langle p^3q\rangle_s}{dt} = -\frac{3}{2}\hbar^2\,a_2 + \frac{\langle p^4\rangle}{m} - 4\,\hbar^2\,a_3\langle q\rangle + 72\,a_3\langle p\rangle^2\langle q\rangle^3 - 18\,a_3\langle p^2\rangle\langle q\rangle^3 - 54\,a_3\langle p\rangle^2\langle q\rangle\langle q^2\rangle + 18\,a_3\langle p^2\rangle\langle q\rangle\langle q^2\rangle + 6\,a_3\langle p\rangle^2\langle q^3\rangle - 3\,a_3\langle p^2\rangle\langle q^3\rangle - 108\,a_3\langle p\rangle\langle q\rangle^2\langle pq\rangle_s + 36\,a_3\langle p\rangle\langle q^2\rangle\langle pq\rangle_s + 36\,a_3\langle q\rangle\langle pq\rangle_s^2 + 18\,a_3\langle q\rangle^2\langle p^2q\rangle_s - 9\,a_3\langle q^2\rangle\langle p^2q\rangle_s + 36\,a_3\langle p\rangle\langle q\rangle\langle pq^2\rangle_s - 18\,a_3\langle pq\rangle_s\langle pq^2\rangle_s - 3\,a_2\langle p^2q^2\rangle_s - 9\,a_3\langle q\rangle\langle p^2q^2\rangle_s - 6\,a_3\langle p\rangle\langle pq^3\rangle_s$
$\frac{d\langle p^4\rangle}{dt} = -4\,\hbar^2\,a_3\langle p\rangle + 96\,a_3\langle p\rangle^3\langle q\rangle^2 - 72\,a_3\langle p\rangle\langle p^2\rangle\langle q\rangle^2 + 8\,a_3\langle p^3\rangle\langle q\rangle^2 - 24\,a_3\langle p\rangle^3\langle q^2\rangle + 24\,a_3\langle p\rangle\langle p^2\rangle\langle q^2\rangle - 4\,a_3\langle p^3\rangle\langle q^2\rangle - 144\,a_3\langle p\rangle^2\langle q\rangle\langle pq\rangle_s + 48\,a_3\langle p^2\rangle\langle q\rangle\langle pq\rangle_s + 48\,a_3\langle p\rangle\langle pq\rangle_s^2 + 48\,a_3\langle p\rangle\langle q\rangle\langle p^2q\rangle_s - 24\,a_3\langle pq\rangle_s\langle p^2q\rangle_s - 4\,a_2\langle p^3q\rangle_s - 8\,a_3\langle q\rangle\langle p^3q\rangle_s + 24\,a_3\langle p\rangle^2\langle pq^2\rangle_s - 12\,a_3\langle p^2\rangle\langle pq^2\rangle_s - 12\,a_3\langle p\rangle\langle p^2q^2\rangle_s$

The products above are symmetrized, for instance $\langle p \cdot q^2\rangle_s = \left\langle \frac{p\cdot q^2 + q^2\cdot p}{2}\right\rangle$

The QHD-4 hierarchy can be shown using the QHDGraphPlot command, compare the table above and the graph below:

```
QHDGraphPlot[mhier4]
```

Closure procedure was applied to order 4



Solution of the QHD-4 Equations

Next we evaluate the hierarchy when the parameters tkes the numerical values used by Pahl and Prezhdo in their paper, and the result of that evaluation is stored in the variable *mnumhier4*:

```
mnumhier4 = mhier4 /. {m → 1, a2 → 1, a3 → 1/2, ℏ → 1};
QHDForm[mnumhier4]
```

Closure procedure was applied to order 4
$\frac{d\langle p\rangle}{dt} = -\langle q\rangle - \frac{\langle q^2\rangle}{2}$
$\frac{d\langle q\rangle}{dt} = \langle p\rangle$
$\frac{d\langle q^2\rangle}{dt} = 2\langle pq\rangle_s$
$\frac{d\langle pq\rangle_s}{dt} = \langle p^2\rangle - \langle q^2\rangle - \frac{\langle q^3\rangle}{2}$
$\frac{d\langle p^2\rangle}{dt} = -2\langle pq\rangle_s - \langle pq^2\rangle_s$
$\frac{d\langle q^3\rangle}{dt} = 3\langle pq^2\rangle_s$
$\frac{d\langle pq^2\rangle_s}{dt} = -\langle q^3\rangle - \frac{\langle q^4\rangle}{2} + 2\langle p^2 q\rangle_s$
$\frac{d\langle q^4\rangle}{dt} = 4\langle pq^3\rangle_s$
$\frac{d\langle p^2 q\rangle_s}{dt} = \langle p^3\rangle - 2\langle pq^2\rangle_s - \langle pq^3\rangle_s$
$\frac{d\langle pq^3\rangle_s}{dt} = \frac{3}{2} + 12\langle q\rangle^5 - 30\langle q\rangle^3\langle q^2\rangle + 15\langle q\rangle\langle q^2\rangle^2 + 10\langle q\rangle^2\langle q^3\rangle - 5\langle q^2\rangle\langle q^3\rangle - \langle q^4\rangle - \frac{5}{2}\langle q\rangle\langle q^4\rangle + 3\langle p^2 q^2\rangle_s$
$\frac{d\langle p^3\rangle}{dt} = -\frac{1}{2} - 3\langle p^2 q\rangle_s - \frac{3}{2}\langle p^2 q^2\rangle_s$
$\frac{d\langle p^2 q^2\rangle_s}{dt} = 24\langle p\rangle\langle q\rangle^4 - 36\langle p\rangle\langle q\rangle^2\langle q^2\rangle + 6\langle p\rangle\langle q^2\rangle^2 + 8\langle p\rangle\langle q\rangle\langle q^3\rangle - \langle p\rangle\langle q^4\rangle - 24\langle q\rangle^3\langle pq\rangle_s + 24\langle q\rangle\langle q^2\rangle\langle pq\rangle_s - 4\langle q^3\rangle\langle pq\rangle_s + 2\langle p^3 q\rangle_s + 12\langle q\rangle^2\langle pq^2\rangle_s - 6\langle q^2\rangle\langle pq^2\rangle_s - 2\langle pq^3\rangle_s - 4\langle q\rangle\langle pq^3\rangle_s$
$\frac{d\langle p^3 q\rangle_s}{dt} = -\frac{3}{2} + \langle p^4\rangle - 2\langle q\rangle + 36\langle p\rangle^2\langle q\rangle^3 - 9\langle p^2\rangle\langle q\rangle^3 - 27\langle p\rangle^2\langle q\rangle\langle q^2\rangle + 9\langle p^2\rangle\langle q\rangle\langle q^2\rangle + 3\langle p\rangle^2\langle q^3\rangle - \frac{3}{2}\langle p^2\rangle\langle q^3\rangle - 54\langle p\rangle\langle q\rangle^2\langle pq\rangle_s + 18\langle p\rangle\langle q^2\rangle\langle pq\rangle_s + 18\langle q\rangle\langle pq\rangle_s^2 + 9\langle q\rangle^2\langle p^2 q\rangle_s - \frac{9}{2}\langle q^2\rangle\langle p^2 q\rangle_s + 18\langle p\rangle\langle q\rangle\langle pq^2\rangle_s - 9\langle pq\rangle_s\langle pq^2\rangle_s - 3\langle p^2 q^2\rangle_s - \frac{9}{2}\langle q\rangle\langle p^2 q^2\rangle_s - 3\langle p\rangle\langle pq^3\rangle_s$
$\frac{d\langle p^4\rangle}{dt} = -2\langle p\rangle + 48\langle p\rangle^3\langle q\rangle^2 - 36\langle p\rangle\langle p^2\rangle\langle q\rangle^2 + 4\langle p^3\rangle\langle q\rangle^2 - 12\langle p\rangle^3\langle q^2\rangle + 12\langle p\rangle\langle p^2\rangle\langle q^2\rangle - 2\langle p^3\rangle\langle q^2\rangle - 72\langle p\rangle^2\langle q\rangle\langle pq\rangle_s + 24\langle p^2\rangle\langle q\rangle\langle pq\rangle_s + 24\langle p\rangle\langle pq\rangle_s^2 + 24\langle p\rangle\langle q\rangle\langle p^2 q\rangle_s - 12\langle pq\rangle_s\langle p^2 q\rangle_s - 4\langle p^3 q\rangle_s - 4\langle q\rangle\langle p^3 q\rangle_s + 12\langle p\rangle^2\langle pq^2\rangle_s - 6\langle p^2\rangle\langle pq^2\rangle_s - 6\langle p\rangle\langle p^2 q^2\rangle_s$

The command QHDInitialConditionsTemplate generates an initial conditions template for the hierarchy:

```
QHDInitialConditionsTemplate[mnumhier4, 0]
```

$$\left\{\langle p\rangle[0] == \blacksquare,\ \langle q\rangle[0] == \blacksquare,\ \left\langle q^2\right\rangle[0] == \blacksquare,\ \langle (p\cdot q)_s\rangle[0] == \blacksquare,\ \left\langle p^2\right\rangle[0] == \blacksquare,\ \left\langle q^3\right\rangle[0] == \blacksquare,\ \left\langle \left(p\cdot q^2\right)_s\right\rangle[0] == \blacksquare,\ \left\langle q^4\right\rangle[0] == \blacksquare,\ \left\langle \left(p^2\cdot q\right)_s\right\rangle[0] == \blacksquare,\ \left\langle \left(p\cdot q^3\right)_s\right\rangle[0] == \blacksquare,\ \left\langle p^3\right\rangle[0] == \blacksquare,\ \left\langle \left(p^2\cdot q^2\right)_s\right\rangle[0] == \blacksquare,\ \left\langle \left(p^3\cdot q\right)_s\right\rangle[0] == \blacksquare,\ \left\langle p^4\right\rangle[0] == \blacksquare\right\}$$

Copy-paste the output of the previous command as input in the next one. Fill in the placeholders (■) with the appropiate initial values. Those initial values can be numbers. On the other hand, in the calculation below they are symbols like *po*, and then these symbols are evaluated at the desired numerical values. These initial conditions correspond to a Gaussian wavepacket, J. Chem Phys. Vol 113 No. 20, May 2002, Pages 8704-8712
http://homepage.cem.itesm.mx/lgomez/quantum/QHDHigherOrders.pdf
The initial conditions are stored in the variable *minicond4* below:

minicond4 =

$$\left\{\langle p\rangle[0] == po,\ \langle q\rangle[0] == qo,\ \left\langle q^2\right\rangle[0] == qo^2 + \frac{\hbar}{2\,m*\omega},\right.$$
$$\langle (p\cdot q)_s\rangle[0] == po*qo,\ \left\langle p^2\right\rangle[0] == po^2 + \frac{\hbar*m*\omega}{2},$$
$$\left\langle q^3\right\rangle[0] == qo^3 + \frac{3}{2}\,qo\,\frac{\hbar}{m*\omega},\ \left\langle \left(p\cdot q^2\right)_s\right\rangle[0] == po*qo^2 + \frac{1}{2}\,po\,\frac{\hbar}{m*\omega},$$
$$\left\langle q^4\right\rangle[0] == qo^4 + 3\,qo^2\,\frac{\hbar}{m*\omega} + \frac{3}{4}\left(\frac{\hbar}{m*\omega}\right)^2,\ \left\langle \left(p^2\cdot q\right)_s\right\rangle[0] == po^2*qo + \frac{1}{2}\,qo*\hbar*m*\omega,$$
$$\left\langle \left(p\cdot q^3\right)_s\right\rangle[0] == po*qo^3 + \frac{3}{2}\,po*qo*\frac{\hbar}{m*\omega},\ \left\langle p^3\right\rangle[0] == po^3 + \frac{3}{2}\,po*\hbar*m*\omega,$$
$$\left\langle \left(p^2\cdot q^2\right)_s\right\rangle[0] == po^2*qo^2 + \frac{1}{2}\,po^2\,\frac{\hbar}{m*\omega} + \frac{qo^2*\hbar*m*\omega}{2} - \frac{\hbar^2}{4},$$
$$\left\langle \left(p^3\cdot q\right)_s\right\rangle[0] == po^3*qo + \frac{3}{2}\,po*qo*\hbar*m*\omega,$$
$$\left.\left\langle p^4\right\rangle[0] == po^4 + 3\,po^2*\hbar*m*\omega + \frac{3}{4}\,(\hbar*m*\omega)^2\right\} \;/.\; \{qo\to 0,\ po\to 0,\ \omega\to 1,\ m\to 1,\ \hbar\to 1\}$$

$$\left\{\langle p\rangle[0] == 0,\ \langle q\rangle[0] == 0,\ \left\langle q^2\right\rangle[0] == \frac{1}{2},\ \langle (p\cdot q)_s\rangle[0] == 0,\ \left\langle p^2\right\rangle[0] == \frac{1}{2},\ \left\langle q^3\right\rangle[0] == 0,\ \left\langle \left(p\cdot q^2\right)_s\right\rangle[0] == 0,\ \left\langle q^4\right\rangle[0] == \frac{3}{4},\ \left\langle \left(p^2\cdot q\right)_s\right\rangle[0] == 0,\ \left\langle \left(p\cdot q^3\right)_s\right\rangle[0] == 0,\ \left\langle p^3\right\rangle[0] == 0,\ \left\langle \left(p^2\cdot q^2\right)_s\right\rangle[0] == -\frac{1}{4},\ \left\langle \left(p^3\cdot q\right)_s\right\rangle[0] == 0,\ \left\langle p^4\right\rangle[0] == \frac{3}{4}\right\}$$

The command QHDNDSolve takes as arguments the numerical version of the hierarchy (which was stored in the variable *mnumhier4*), the initial conditions (which were stored in *minicond4*), the initial time and the final time. The output of this command is the numerical solution of the differential equations, in the form of InterpolatingFunction objects. This output can be used to plot (graph) the dynamical variables as functions of time, as it will be shown below in this document. The output is stored in the variable *msol4* in the calculation below:

```
msol4 = QHDNDSolve[mnumhier4, minicond4, 0, 4]
```

```
{{⟨p⟩ → InterpolatingFunction[{{0., 4.}}, <>],
  ⟨q⟩ → InterpolatingFunction[{{0., 4.}}, <>],
  ⟨q^2⟩ → InterpolatingFunction[{{0., 4.}}, <>],
  ⟨(p · q)_s⟩ → InterpolatingFunction[{{0., 4.}}, <>],
  ⟨p^2⟩ → InterpolatingFunction[{{0., 4.}}, <>],
  ⟨q^3⟩ → InterpolatingFunction[{{0., 4.}}, <>],
  ⟨(p · q^2)_s⟩ → InterpolatingFunction[{{0., 4.}}, <>],
  ⟨q^4⟩ → InterpolatingFunction[{{0., 4.}}, <>],
  ⟨(p^2 · q)_s⟩ → InterpolatingFunction[{{0., 4.}}, <>],
  ⟨(p · q^3)_s⟩ → InterpolatingFunction[{{0., 4.}}, <>],
  ⟨p^3⟩ → InterpolatingFunction[{{0., 4.}}, <>],
  ⟨(p^2 · q^2)_s⟩ → InterpolatingFunction[{{0., 4.}}, <>],
  ⟨(p^3 · q)_s⟩ → InterpolatingFunction[{{0., 4.}}, <>],
  ⟨p^4⟩ → InterpolatingFunction[{{0., 4.}}, <>]}}
```

The command command QHDPlot takes as its second argument the output of QHDNDSolve, which was stored in the variable *msol4*. The first argument specifies the dynamical variables or expression that we want to plot (graph) as a function of time, see the plot below:

```
QHDPlot[q, msol4]
```



The command command QHDPlot takes as its second argument the output of QHDNDSolve, which was stored in the variable *msol4*. The first argument specifies the dynamical variables or expression that we want to plot (graph) as a function of time, see the plot below:

```
mgraph4 = QHDPlot[q, msol4, PlotStyle → Darker[Orange]]
```



The standard *Mathematica* command Show can be used to compare *mgraph4*, which was stored above, with *mgraph3* and *mgraph2*, which were calculated in previous sections. The paper by Pahl and Prezhdo shows that QHD-4 is a better approximation to Quantum Mechanics than QHD-3 and QHD-2, see the graph below:

```
Show[mgraph2, mgraph3, mgraph4]
```



Standard *Mathematica* options can be used so that the graph looks like the corresponding graph in Pahl and Prezhdo paper, see the result below:

```
Show[mgraph2, mgraph3, mgraph4,
 Axes → True, AxesOrigin → {0, -2}, PlotLabel →
  "Decay from Metastable Potential\n Compare with Pahl and Prezhdo\n
    J.Chem.Phys., Vol 116, No. 20, May 2002\n Pages 8704-8712 Fig 3"]
```



Saving the QHD hierarchies for future use

The calculation QHD hierarchies is time consuming, therefore it is a good idea to save them for future use. The simplest way to do it is using the standard *Mathematica* command Put. The QHD-2 hierarchy is stored in the file qhd2.m, see the command below:

```
Put[mhier2, "qhd2.m"]
```

The QHD-3 hierarchy is stored in the file qhd3.m, see the command below:

```
Put[mhier3, "qhd3.m"]
```

The QHD-4 hierarchy is stored in the file qhd4.m, see the command below:

```
Put[mhier4, "qhd4.m"]
```

The command FileNames[“*.m”] can be used to see the files that were created by the command Put (together with any other preexisting file with .m extension), see the command below:

```
FileNames["*.m"]
```

```
{hier4.m, HierarchyWater.m, qhd2.m, qhd3.m, qhd4.m, schrodingersolbarrier.m}
```

Dependence of Decay Time on $\hbar$

Next commands have been written as they would be used in a different *Mathematica* session, using the command Get to load the hierarchy that was saved above (instead of calculating it again). Notice that the commutation relation and the Hamiltonian are defined again, and its important to use those that correspond to the hierarchy that is loaded. Next commands calculate and graph the dependence of decay time on $\hbar$, the result is stored in the variable data2, which will be used later to reproduce a graph of the paper by Pahl and Prezhdo, see the graph below:

```
Needs["Quantum`QHD`"];
SetQuantumObject[q, p];
〚q, p〛₋ = ⅈ * ℏ;
mh = p^2/(2 m) + a2/2 q^2 + a3/3 q^3;
hhier2 = Get["qhd2.m"];
iniguess = 15;
data2 = Table[
   hnumhier2 = hhier2 /. {m → 1, a2 → 1, a3 → 1/2, ℏ → h};
   hinicond2 =
    {⟨p⟩[0] == po, ⟨q⟩[0] == qo, ⟨q^2⟩[0] == qo^2 + ℏ/(2 m * ω), ⟨(p · q)s⟩[0] == po * qo,
      ⟨p^2⟩[0] == po^2 + (ℏ * m * ω)/2} /. {qo → 0, po → 0, ω → 1, m → 1, ℏ → h};
   hsol2 = QHDNDSolve[hnumhier2, hinicond2, 0, iniguess + 0.5];
   hqf2 = QHDFunction[q, hsol2];
   htime = (ht /. FindRoot[hqf2[ht] == -2, {ht, iniguess},
        AccuracyGoal → 3, PrecisionGoal → 3, MaxIterations → 1000]);
   iniguess = htime;
   {h, htime},
   {h, 0.85, 2, 0.025}];
ListLinePlot[data2, AxesLabel -> {Style[ℏ, Large], "crossing time τ"}]
```



Below we have the same calculation as above, but with QHD-3 instead of QHD-2, using the “qhd3.m” file, and storing the result in data3:

```
Needs["Quantum`QHD`"];
SetQuantumObject[q, p];
〚q, p〛₋ = ⅈ * ℏ;
mh = p^2/(2 m) + a2/2 q^2 + a3/3 q^3;
hhier3 = Get["qhd3.m"];
iniguess = 8;
data3 = Table[
   hnumhier3 = hhier3 /. {m → 1, a2 → 1, a3 → 1/2, ℏ → h};
   hinicond3 = {⟨p⟩[0] == po, ⟨q⟩[0] == qo, ⟨q^2⟩[0] == qo^2 + ℏ/(2 m * ω),
       ⟨(p · q)_s⟩[0] == po * qo, ⟨p^2⟩[0] == po^2 + (ℏ * m * ω)/2, ⟨q^3⟩[0] == qo^3 + 3/2 qo ℏ/(m * ω),
       ⟨(p · q^2)_s⟩[0] == po * qo^2 + 1/2 po ℏ/(m * ω), ⟨(p^2 · q)_s⟩[0] == po^2 * qo + 1/2 qo * ℏ * m * ω,
       ⟨p^3⟩[0] == po^3 + 3/2 po * ℏ * m * ω} /. {qo → 0, po → 0, ω → 1, m → 1, ℏ → h};
   hsol3 = QHDNDSolve[hnumhier3, hinicond3, 0, iniguess + 0.5];
   hqf3 = QHDFunction[q, hsol3];
   htime = (ht /. FindRoot[hqf3[ht] == -2, {ht, iniguess},
        AccuracyGoal → 3, PrecisionGoal → 3, MaxIterations → 1000]);
   iniguess = htime;
   {h, htime},
   {h, 0.6, 2, 0.025}];
ListLinePlot[data3, AxesLabel -> {Style[ℏ, Large], "crossing time τ"}]
```



Below we have the same calculation as above, but with QHD-4 instead of QHD-3 or QHD-2, using the “qhd4.m” file, and storing the result in data4:

```
Needs["Quantum`QHD`"];
SetQuantumObject[q, p];
⟦q, p⟧₋ = ⅈ * ℏ;
mh = p^2/(2 m) + a₂/2 q^2 + a₃/3 q^3;
hhier4 = Get["qhd4.m"];
iniguess = 6;
data4 = Table[
   hnumhier4 = hhier4 /. {m → 1, a₂ → 1, a₃ → 1/2, ℏ → h};
   hinicond4 = {⟨p⟩[0] == po, ⟨q⟩[0] == qo,
      ⟨q^2⟩[0] == qo^2 + ℏ/(2 m * ω), ⟨(p · q)_s⟩[0] == po * qo, ⟨p^2⟩[0] == po^2 + (ℏ * m * ω)/2,
      ⟨q^3⟩[0] == qo^3 + 3/2 qo ℏ/(m * ω), ⟨(p · q^2)_s⟩[0] == po * qo^2 + 1/2 po ℏ/(m * ω),
      ⟨q^4⟩[0] == qo^4 + 3 qo^2 ℏ/(m * ω) + 3/4 (ℏ/(m * ω))^2, ⟨(p^2 · q)_s⟩[0] == po^2 * qo + 1/2 qo * ℏ * m * ω,
      ⟨(p · q^3)_s⟩[0] == po * qo^3 + 3/2 po * qo * ℏ/(m * ω), ⟨p^3⟩[0] == po^3 + 3/2 po * ℏ * m * ω,
      ⟨(p^2 · q^2)_s⟩[0] == po^2 * qo^2 + 1/2 po^2 ℏ/(m * ω) + (qo^2 * ℏ * m * ω)/2 - ℏ^2/4,
      ⟨(p^3 · q)_s⟩[0] == po^3 * qo + 3/2 po * qo * ℏ * m * ω, ⟨p^4⟩[0] ==
       po^4 + 3 po^2 * ℏ * m * ω + 3/4 (ℏ * m * ω)^2} /. {qo → 0, po → 0, ω → 1, m → 1, ℏ → h};
   hsol4 = QHDNDSolve[hnumhier4, hinicond4, 0, iniguess + 0.5];
   hqf4 = QHDFunction[q, hsol4];
   htime = (ht /. FindRoot[hqf4[ht] == -2, {ht, iniguess},
        AccuracyGoal → 3, PrecisionGoal → 3, MaxIterations → 1000]);
   iniguess = htime;
   {h, htime},
   {h, 0.6, 2, 0.025}];
ListLinePlot[data4, AxesLabel -> {Style[ℏ, Large], "crossing time τ"}]
```



Standard *Mathematica* options can be used so that the graph looks like the corresponding graph in Pahl and Prezhdo paper, see the result below:

```
ListLinePlot[{data2, data3, data4}, Frame → True,
 FrameLabel → {Style[ℏ, Large], "crossing time τ"},
 PlotRange → {{0.6, 2}, {2, 10}}, PlotLabel →
  "Decay from Metastable Potential\n Compare with Pahl and Prezhdo\n
    J.Chem.Phys., Vol 116, No. 20, May 2002\n Pages 8704-8712 Fig 2"]
```



Dependence of Decay Time on po

Next commands have been written as they would be used in a different *Mathematica* session, using the command Get to load the hierarchy that was saved above (instead of calculating it again). Notice that the commutation relation and the Hamiltonian are defined again, and its important to use those that correspond to the hierarchy that is loaded. Next commands calculate and graph the dependence of decay time on p_0, the initial momentum, the result is used to reproduce part of a graph of the paper by Pahl and Prezhdo, see the graph below:

```
Needs["Quantum`QHD`"];
SetQuantumObject[q, p];
⟦q, p⟧₋ = ⅈ * ℏ;
mh = p^2/(2 m) + a2/2 q^2 + a3/3 q^3;
hhier2 = Get["qhd2.m"];
hnumhier2 = hhier2 /. {m → 1, a2 → 1, a3 → 1/2, ℏ → 1};
iniguess = 1;
pdata2 = Table[
   hinicond2 =
    {⟨p⟩[0] == po, ⟨q⟩[0] == qo, ⟨q^2⟩[0] == qo^2 + ℏ/(2 m * ω), ⟨(p · q)s⟩[0] == po * qo,
      ⟨p^2⟩[0] == po^2 + (ℏ * m * ω)/2} /. {qo → 0, ω → 1, m → 1, ℏ → 1};
   hsol2 = QHDNDSolve[hnumhier2, hinicond2, 0, iniguess + 0.5];
   hqf2 = QHDFunction[q, hsol2];
   htime = (ht /. FindRoot[hqf2[ht] == -2, {ht, iniguess},
        AccuracyGoal → 3, PrecisionGoal → 3, MaxIterations → 1000]);
   iniguess = htime;
   {po, htime},
   {po, -2, 5, 0.05}];
ListLinePlot[pdata2, Frame → True, Axes → False,
 FrameLabel → {Style[p0, Large], "crossing time τ"},
 PlotLabel → "Decay from Metastable
    Potential\nCompare with Pahl and Prezhdo\nJ.Chem.Phys.,
    Vol 116, No. 20, May 2002\nPages 8704-8712 Fig 4(b)"]
```



This tutorial shows how to use the QUANTUM *Mathematica* add-on to reproduce the results and graphs from Pahl and Prezhdo in J. Chem Phys. Vol 116 No. 20, May 2002, Pages 8704-8712
http://homepage.cem.itesm.mx/lgomez/quantum/QHDHigherOrders.pdf

---

by José Luis Gómez-Muñoz
http://homepage.cem.itesm.mx/lgomez/quantum/
jose.luis.gomez@itesm.mx