QHD for a general potential

by José Luis Gómez-Muñoz
http://homepage.cem.itesm.mx/lgomez/quantum/
jose.luis.gomez@itesm.mx

Introduction

Quantized Hamilton Dynamics (QHD) is applied to the **a general potential**. QHD gives an approximation to the Heisenberg Equations of Motion (EOM). The QHD commands used in this document are included in QUANTUM, which is a free *Mathematica* add-on that can be downloaded from
http://homepage.cem.itesm.mx/lgomez/quantum/
This tutorial shows how to use the QUANTUM *Mathematica* add-on to reproduce the results and graphs from Prezhdo and Pereverzev in J. Chem. Phys., Vol 116, No. 11, March 2002, pages 4450-4461
http://homepage.cem.itesm.mx/lgomez/quantum/QHDgeneralpotential.pdf .

Load the Package

First load the Quantum`QHD` package. Write:
Needs["Quantum`QHD`"]
then press at the same time the keys SHIFT-ENTER to evaluate. *Mathematica* will load the package and print a welcome message:

```
Needs["Quantum`QHD`"]
```

```
Quantum`QHD`
A Mathematica package for Quantized Hamilton
  Dynamics approximation to Heisenberg Equations of Motion
by José Luis Gómez-Muñoz
based on the original idea of Kirill Igumenshchev

This add-on does NOT work properly with the debugger turned on. Therefore
  the debugger must NOT be checked in the Evaluation menu of Mathematica.

Execute SetQHDAliases[] in order to use the keyboard to enter QHD objects
SetQHDAliases[] must be executed again in each new notebook that is created
```

In order to use the keyboard to enter quantum objects write:
SetQHDAliases[ ];
then press at the same time the keys SHIFT-ENTER to evaluate. Remember that SetQuantumAliases[ ] must be evaluated again in each new notebook:

```
SetQHDAliases[]
```

```
ALIASES:
[ESC]on[ESC]      · Quantum concatenation symbol
[ESC]time[ESC]    t Time symbol
[ESC]hb[ESC]      ℏ Reduced Planck's constant (h bar)
[ESC]ii[ESC]      ⅈ Imaginary I symbol
[ESC]inf[ESC]     ∞ Infinity symbol
[ESC]->[ESC]      → Option (Rule) symbol
[ESC]ave[ESC]     〈□〉 Quantum average template
[ESC]expec[ESC]   〈□〉 Quantum average template
[ESC]symm[ESC]    (□·□)_s Symmetrized quantum product template
[ESC]comm[ESC]    ⟦□,□⟧_- Commutator template
[ESC]po[ESC]      (□)^□ Power template
[ESC]su[ESC]      □_□ Subscripted variable template
[ESC]posu[ESC]    □_□^□ Power of a subscripted variable template
[ESC]fra[ESC]     □/□ Fraction template
[ESC]eva[ESC]     □/.{□→□,□→□} Evaluation (ReplaceAll) template

SetQHDAliases[] must be executed again in each
  new notebook that is created, only one time per notebook.
```

Commutation Relationship

Here we define the commutation relationship that will be used in this document.
In order to enter the templates and symbols $[\![\square, \square]\!]_-$, $\hbar$ and i you can either use the QHD palette (toolbar) or press the keys [ESC]comm[ESC], [ESC]hb[ESC] and [ESC]ii[ESC]

```
Clear[q, p];
SetQuantumObject[q, p];
⟦q, p⟧_- = ⅈ * ℏ
```

```
ⅈ ℏ
```

Moving Frame Approximation to the Potential Energy

The potential energy is expanded in Taylor series around the instantaneous average value of the position variable $\langle \mathbf{q} \rangle$. This is the "moving frame approximation". Please remember that the command SetQuantumObject was used above in order to define **q** as a quantum operator. The fourth order moving frame approximation to the potential energy is stored in the variable *v4* below:

```
v4 = Sum[D[v[⟨q⟩], {⟨q⟩, j}]/j! (q - ⟨q⟩)^j, {j, 0, 4}]
```

$$v[\langle q\rangle] + (q-\langle q\rangle)\, v'[\langle q\rangle] + \frac{1}{2}(q-\langle q\rangle)^2\, v''[\langle q\rangle] + \frac{1}{6}(q-\langle q\rangle)^3\, v^{(3)}[\langle q\rangle] + \frac{1}{24}(q-\langle q\rangle)^4\, v^{(4)}[\langle q\rangle]$$

Using a unitary mass, the Hamiltonian is stored in *h4* below:

```
h4 = p^2/2 + v4
```

$$\frac{p^2}{2} + v[\langle q\rangle] + (q-\langle q\rangle)\, v'[\langle q\rangle] + \frac{1}{2}(q-\langle q\rangle)^2\, v''[\langle q\rangle] + \frac{1}{6}(q-\langle q\rangle)^3\, v^{(3)}[\langle q\rangle] + \frac{1}{24}(q-\langle q\rangle)^4\, v^{(4)}[\langle q\rangle]$$

Second-Order QHD with a Fourth-Order Moving-Frame Expansion of the Potential

The evolution of the average of an observable *A* in the Heisenberg representation is given by the equation of motion (EOM):

$$i\hbar \frac{d}{dt}\langle A\rangle = \langle [A, H]\rangle$$

Consider the averages of momentum, position and their products $\langle p\rangle$, $\langle q\rangle$, $\langle p^2\rangle$, $\langle q^2\rangle$, $\langle pq\rangle$, $\langle p^3\rangle$, $\langle p^2 q\rangle$... The EOMs for the average values are coupled and, in general, form an infinite hierarchy of equations. Quantized Hamilton Dynamics (QHD) terminates this hierarchy by the approximation of the higher order averages via products of the lower order averages. For instance the approximation

$$\langle ABC\rangle \approx \langle AB\rangle\langle C\rangle + \langle AC\rangle\langle B\rangle + \langle BC\rangle\langle A\rangle - 2\langle A\rangle\langle B\rangle\langle C\rangle$$

can be used to approximate the third-order averages $\langle q^3\rangle$ and $\langle pq^2\rangle_s$ in terms of the first and second order averages $\langle p\rangle$, $\langle q\rangle$, $\langle p^2\rangle$, $\langle q^2\rangle$ and $\langle p\cdot q\rangle_s = \left\langle \frac{p\cdot q + q\cdot p}{2}\right\rangle$.

The command QHDHierarchy (see below) takes as its first argument the QHD order, the second argument is the variable which is used to start the hierarchy, and the third argument is the Hamiltonian. The resulting hierarchy is stored in the variable *hier4* and it is shown in a nice format using the command QHDForm below:

```
hier4 = QHDHierarchy[2, q, h4];
QHDForm[hier4,
 QHDLabel → "Compare with Prezhdo and Pereverzev \nJ.Chem.Phys. Vol
    116 No 11, March 2002\nPages 4450-4461 Eqs.(15)-(19)"]
```

Compare with Prezhdo and Pereverzev
J.Chem.Phys. Vol 116 No 11, March 2002
Pages 4450-4461 Eqs.(15)-(19)

$$\frac{d\langle q\rangle}{dt} = \langle p\rangle$$

$$\frac{d\langle p\rangle}{dt} = -v'(\langle q\rangle) + \frac{1}{2}\langle q\rangle^2 v^{(3)}(\langle q\rangle) - \frac{1}{2}\left\langle q^2\right\rangle v^{(3)}(\langle q\rangle)$$

$$\frac{d\langle q^2\rangle}{dt} = 2\langle pq\rangle_s$$

$$\frac{d\langle pq\rangle_s}{dt} = \left\langle p^2\right\rangle - \langle q\rangle v'(\langle q\rangle) + \langle q\rangle^2 v''(\langle q\rangle) - \left\langle q^2\right\rangle v''(\langle q\rangle) + \frac{1}{2}\langle q\rangle^3 v^{(3)}(\langle q\rangle) - \frac{1}{2}\langle q\rangle\left\langle q^2\right\rangle v^{(3)}(\langle q\rangle) - \frac{1}{2}\langle q\rangle^4 v^{(4)}(\langle q\rangle) + \langle q\rangle^2\left\langle q^2\right\rangle v^{(4)}(\langle q\rangle) - \frac{1}{2}\left\langle q^2\right\rangle^2 v^{(4)}(\langle q\rangle)$$

$$\frac{d\langle p^2\rangle}{dt} = -2\langle p\rangle v'(\langle q\rangle) + 2\langle p\rangle\langle q\rangle v''(\langle q\rangle) - 2\langle pq\rangle_s v''(\langle q\rangle) + \langle p\rangle\langle q\rangle^2 v^{(3)}(\langle q\rangle) - \langle p\rangle\left\langle q^2\right\rangle v^{(3)}(\langle q\rangle) - \langle p\rangle\langle q\rangle^3 v^{(4)}(\langle q\rangle) + \langle p\rangle\langle q\rangle\left\langle q^2\right\rangle v^{(4)}(\langle q\rangle) + \langle q\rangle^2\langle pq\rangle_s v^{(4)}(\langle q\rangle) - \left\langle q^2\right\rangle\langle pq\rangle_s v^{(4)}(\langle q\rangle)$$

Example: Double Well with a Third-Order Moving-Frame Expansion of the Potential

The potential energy that will be used in this example is
$V = -5\,q^2 + q^4$
It is used by Prezhdo and Pereverzev in their paper J. Chem. Phys., Vol 116, No. 11, March 2002, pages 4450-4461
http://homepage.cem.itesm.mx/lgomez/quantum/QHDgeneralpotential.pdf, this is a “double well”potential energy, see the graph below:

```
Plot[-5 q^2 + q^4, {q, -3, 3}]
```



This is a third-order moving frame approximation to the potential, see below:

```
Clear[a, b];
vd[q_] := a * q^2 + b * q^4;
```

$$vd3 = \sum_{j=0}^{3} \frac{D[\, vd[\langle q\rangle],\ \{\langle q\rangle,\ j\}\,]}{j\,!}\ (q - \langle q\rangle)^{j}$$

$$4\,b\,(q - \langle q\rangle)^3 \langle q\rangle + a \langle q\rangle^2 + b \langle q\rangle^4 + \frac{1}{2}\,(q - \langle q\rangle)^2 \left(2\,a + 12\,b\,\langle q\rangle^2\right) + (q - \langle q\rangle)\,\left(2\,a\,\langle q\rangle + 4\,b\,\langle q\rangle^3\right)$$

Using a unitary mass, the Hamiltonian is stored in *hd3* below:

$$hd3 = \frac{p^2}{2} + vd3$$

$$\frac{p^2}{2} + 4\,b\,(q - \langle q\rangle)^3 \langle q\rangle + a \langle q\rangle^2 + b \langle q\rangle^4 + \frac{1}{2}\,(q - \langle q\rangle)^2 \left(2\,a + 12\,b\,\langle q\rangle^2\right) + (q - \langle q\rangle)\,\left(2\,a\,\langle q\rangle + 4\,b\,\langle q\rangle^3\right)$$

The commands below calculate and show the second order QHD hierarchy of equations for the third order moving frame approximation to the double-well. The command QHDHierarchy takes as its first argument the QHD order, the second argument is the variable which is used to start the hierarchy, and the third argument is the Hamiltonian. The resulting hierarchy is stored in the variable *hierd3* and it is shown in a nice format using the command QHDForm below:

```
hierd3 = QHDHierarchy[2, q, hd3];
QHDForm[hierd3]
```

Closure procedure was applied to order 2
$\frac{d\langle q\rangle}{dt} = \langle p\rangle$
$\frac{d\langle p\rangle}{dt} = -2\,a\,\langle q\rangle + 8\,b\,\langle q\rangle^3 - 12\,b\,\langle q\rangle\,\langle q^2\rangle$
$\frac{d\langle q^2\rangle}{dt} = 2\,\langle pq\rangle_s$
$\frac{d\langle pq\rangle_s}{dt} = \langle p^2\rangle + 20\,b\,\langle q\rangle^4 - 2\,a\,\langle q^2\rangle - 24\,b\,\langle q\rangle^2\,\langle q^2\rangle$
$\frac{d\langle p^2\rangle}{dt} = 40\,b\,\langle p\rangle\,\langle q\rangle^3 - 24\,b\,\langle p\rangle\,\langle q\rangle\,\langle q^2\rangle - 4\,a\,\langle pq\rangle_s - 24\,b\,\langle q\rangle^2\,\langle pq\rangle_s$

A QHD hierarchy can be shown in a graph using the command QHDGraphPlot on the output of QHDHierachy. Arrows point from a first dynamical variable to a second dynamical variable that includes the first one in its EOM; please compare the table above with the graph below:

```
QHDGraphPlot[hierd3]
```

Closure procedure was applied to order 2



Next we evaluate the hierarchy when the parameter *a* takes the value of -5 and *b* takes the value of 1, and the result of that evaluation is stored in the variable *numhierd3*:

```
numhierd3 = hierd3 /. {a → -5, b → 1};
QHDForm[ numhierd3 ]
```

Closure procedure was applied to order 2
$\frac{d\langle q\rangle}{dt} = \langle p\rangle$
$\frac{d\langle p\rangle}{dt} = 10\,\langle q\rangle + 8\,\langle q\rangle^3 - 12\,\langle q\rangle\,\langle q^2\rangle$
$\frac{d\langle q^2\rangle}{dt} = 2\,\langle pq\rangle_s$
$\frac{d\langle pq\rangle_s}{dt} = \langle p^2\rangle + 20\,\langle q\rangle^4 + 10\,\langle q^2\rangle - 24\,\langle q\rangle^2\,\langle q^2\rangle$
$\frac{d\langle p^2\rangle}{dt} = 40\,\langle p\rangle\,\langle q\rangle^3 - 24\,\langle p\rangle\,\langle q\rangle\,\langle q^2\rangle + 20\,\langle pq\rangle_s - 24\,\langle q\rangle^2\,\langle pq\rangle_s$

The command QHDInitialConditionsTemplate generates an initial conditions template for the hierarchy:

```
QHDInitialConditionsTemplate[numhierd3 , 0]
```

$$\{\langle q\rangle[0] == ■,\ \langle p\rangle[0] == ■,\ \langle q^2\rangle[0] == ■,\ \langle (p\cdot q)_s\rangle[0] == ■,\ \langle p^2\rangle[0] == ■\}$$

Copy-paste the output of the previous command as input in the next one. Fill in the placeholders (■) with the appropiate initial values. Those initial values can be numbers. On the other hand, in the calculation below they are symbols like *po*, and then these symbols are evaluated at the desired numerical values. The initial conditions are stored in the variable *inicond3* below:

$$\text{inicond3} = \left\{\langle q\rangle[0] == qo,\ \langle p\rangle[0] == po,\ \langle q^2\rangle[0] == qo^2 + \frac{\hbar}{2\,\omega},\ \langle (p\cdot q)_s\rangle[0] == po * qo,\ \langle p^2\rangle[0] == po^2 + \frac{\hbar * \omega}{2}\right\} \text{ /. } \left\{\hbar \to 1,\ po \to 0,\ qo \to -\sqrt{\frac{5}{2}},\ \omega \to \sqrt{-4\,a}\right\} \text{ /. } \{a \to -5\}$$

$$\left\{\langle q\rangle[0] == -\sqrt{\frac{5}{2}},\ \langle p\rangle[0] == 0,\ \langle q^2\rangle[0] == \frac{5}{2} + \frac{1}{4\sqrt{5}},\ \langle (p\cdot q)_s\rangle[0] == 0,\ \langle p^2\rangle[0] == \sqrt{5}\right\}$$

The command QHDNDSolve takes as arguments the numerical version of the hierarchy (which was stored in the variable *numhierd3*), the initial conditions (which were stored in *inicond3*), the initial time and the final time. The output of this command is the numerical solution of the differential equations, in the form of InterpolatingFunction objects. This output can be used to plot (graph) the dynamical variables as functions of time, as it will be shown below in this document. The output is stored in the variable *sold3* in the calculation below:

```
sold3 = QHDNDSolve[numhierd3, inicond3, 0, 3]
```

```
{{⟨q⟩ → InterpolatingFunction[{{0., 3.}}, <>],
  ⟨p⟩ → InterpolatingFunction[{{0., 3.}}, <>],
  ⟨q^2⟩ → InterpolatingFunction[{{0., 3.}}, <>],
  ⟨(p · q)_s⟩ → InterpolatingFunction[{{0., 3.}}, <>],
  ⟨p^2⟩ → InterpolatingFunction[{{0., 3.}}, <>]}}
```

The command command QHDPlot takes as its second argument the output of QHDNDSolve, which was stored in the variable *sold3*. The first argument specifies the dynamical variables or expression that we want to plot (graph) as a function of time. If we want to plot all the dynamical variables, the first argument must be the word *All*, see the five plots below:

```
QHDPlot[All, sold3]
```





−0.2
−0.4
0.0
0.5
1.0
1.5
2.0
2.5
3.0
t
⟨q²⟩
2.6
2.5
2.4
2.3
2.2
2.1
2.0
1.9
0.0
0.5
1.0
1.5
2.0
2.5
3.0
t
⟨pq⟩s
0.6
0.4
0.2
0.0
−0.2
−0.4
−0.6



Next command generates the plot of $\langle$q$\rangle$ as a function of time. It reproduces part of a figure of the paper by Prezhdo and Pereverzev, see the graph below:

```
QHDPlot[q, sold3,
 PlotRange → {-1.7, -1.2},
 AxesOrigin → {0, -Sqrt[5/2]}, Axes → True,
 PlotLabel →
  "Compare with Prezhdo and Pereverzev \nJ.Chem.Phys. Vol 116 No 11,
    March 2002\nPages 4450-4461 Fig. 2(a)"
]
```



A different set of initial conditions is stored in *inicond3b* below:

$$\text{inicond3b} = \left\{\langle q\rangle[0] == qo,\ \langle p\rangle[0] == po,\ \langle q^2\rangle[0] == qo^2 + \frac{\hbar}{2\,\omega},\ \langle (p\cdot q)_s\rangle[0] == po * qo,\ \langle p^2\rangle[0] == po^2 + \frac{\hbar * \omega}{2}\right\} /.\ \left\{\hbar \to 1,\ po \to 0,\ qo \to \frac{-22}{10},\ \omega \to \sqrt{-4\,a}\right\} /.\ \{a \to -5\}$$

$$\left\{\langle q\rangle[0] == -\frac{11}{5},\ \langle p\rangle[0] == 0,\ \langle q^2\rangle[0] == \frac{121}{25} + \frac{1}{4\sqrt{5}},\ \langle (p\cdot q)_s\rangle[0] == 0,\ \langle p^2\rangle[0] == \sqrt{5}\right\}$$

Solution of the differential equations with the new initial conditions:

```
sold3b = QHDNDSolve[numhierd3, inicond3b, 0, 1.5]
```

```
{{⟨q⟩ → InterpolatingFunction[{{0., 1.5}}, <>],
  ⟨p⟩ → InterpolatingFunction[{{0., 1.5}}, <>],
  ⟨q²⟩ → InterpolatingFunction[{{0., 1.5}}, <>],
  ⟨(p · q)_s⟩ → InterpolatingFunction[{{0., 1.5}}, <>],
  ⟨p²⟩ → InterpolatingFunction[{{0., 1.5}}, <>]}}
```

Next command generates the plot of $\langle q \rangle$ as a function of time. It reproduces part of a figure of the paper by Prezhdo and Pereverzev, it shows a particle initially localize in the left minimum on the potential, and in the long run becoming equally split between the two wells, see the graph below:

```
QHDPlot[q, sold3b,
 PlotRange → {-3, 2},
 AxesOrigin → {0, 0}, Axes → True,
 PlotLabel →
  "Compare with Prezhdo and Pereverzev \nJ.Chem.Phys. Vol 116 No 11,
    March 2002\nPages 4450-4461 Fig. 2(b)"
]
```



This tutorial shows how to use the QUANTUM *Mathematica* add-on to reproduce the results and graphs from Prezhdo and Pereverzev in J. Chem. Phys., Vol 116, No. 11, March 2002, pages 4450-4461
http://homepage.cem.itesm.mx/lgomez/quantum/QHDgeneralpotential.pdf .

by José Luis Gómez-Muñoz

http://homepage.cem.itesm.mx/lgomez/quantum/
jose.luis.gomez@itesm.mx